/*

Jessica Calcaro	
Proposal Title: Public Perceptions of Prenatal Alcohol Consumption	


HYPOTHESES

Stated-Hyp1: Public opinion toward pregnant women’s engagement in potentially risky behaviors varies by the severity of the risk.

	Test-Hyp1: Disapproval towards pregnant women's alcohol consumption will vary by degree of consumption (i.e. coefficient of no vs. light drinking will be smaller than coefficient of no vs. heavy drinking - see page 1 of proposal)

Stated-Hyp2: Public opinion toward pregnant women’s engagement in potentially risky behaviors varies by the pregnant women’s social class.

	Test-Hyp2: Disapproval towards pregnant women's alcohol consumption will vary by women's social status (high vs. low)
	
Stated-Hyp3: Public opinion toward pregnant women’s engagement in potentially risky behaviors varies by signaling of the medical community’s recommendations.

	Test-Hyp3: Disapproval towards pregnant women's alcohol consumption will vary by if signal about doctor's warning is included.

Stated-Hyp4: Public opinion toward pregnant women’s engagement in potentially risky behaviors varies by the characteristics of those providing their views. (see bullet # 2d on page 1 of proposal; and first highlighted sentence on page 2)
	
	Test-Hyp4: Disapproval towards pregnant women's alcohol consumption will be higher among less advantaged respondents.
	
	Test-Hyp5: Disapproval towards pregnant women's alcohol consumption will vary among men and women respondents.
	
********************************************************************************
NOTES:

- We are basing the interaction hypotheses on the bulleted points on first page 
of the proposal.	

- The outcomes are confusing in the control condition of no drinking. (i.e. do 
you disapprove of megan only drinking water? do you think megan's drinking 
water will harm her baby?) 

*/

clear all
use CalarcoS38.dta, clear

********************************************************************************


*MEASURES OF EXPERIMENTAL MANIPULATION

tab CALARCO_VIGNETTE
/* vignettes: from codebook
1	HIGH STATUS, NO DRINKING, NO WARNING
2	HIGH STATUS, NO DRINKING, WARNING
3	HIGH STATUS, LIGHT DRINKING, NO WARNING
4	HIGH STATUS, LIGHT DRINKING, WARNING
5	HIGH STATUS, HEAVY DRINKING, NO WARNING
6	HIGH STATUS, HEAVY DRINKIG, WARNING
7	LOW STATUS, NO DRINKING, NO WARNING
8	LOW STATUS, NO DRINKING, WARNING
9	LOW STATUS, LIGHT DRINKING, NO WARNING
10	LOW STATUS, LIGHT DRINKING, WARNING
11	LOW STATUS, HEAVY DRINKING, NO WARNING
12	LOW STATUS, HEAVY DRINKING, WARNING
*/

* severity of drinking
	lab def severity 0 "none" 1 "light drinking" 2 "binge drinking"
	recode CALARCO_VIGNETTE (1/2=0) (3/4=1) (5/6=2) (7/8=0) (9/10=1) (11/12=2) (*=.), gen(severity_drinking)
	lab val severity_drinking severity
	tab severity_drinking

* any drinking (creating this variable to test the remaining hypotheses)
	gen anydrinking = 1 if severity_drinking!=0
	replace anydrinking=0 if severity_drinking==0
	
* woman's social status
	gen status_low=1 if CALARCO_VIGNETTE>=7
	replace status_low= 0 if CALARCO_VIGNETTE<=6
	tab status_low

* doctor's warning signalled
	recode CALARCO_VIGNETTE (1=0) (2=1) (3=0) (4=1) (5=0) (6=1) (7=0) (8=1) (9=0) (10=1) (11=0) (12=1) , gen(doctorwarning)
	tab doctorwarning

* OUTCOMES

* To what extent do you approve or disapprove of Megan’s decision to [drink only water/sip one glass of wine/finish two glasses of wine] at the bar?
	tab Q7
	replace Q7=. if Q7>5 // recode missing

* Do you think that Megan’s decision to [drink only water/sip one glass of wine/finish two glasses of wine] will cause harm to her baby?	
	tab Q8
	replace Q8=. if Q8>5 // recode missing
	
	* create summated measure of disapproval
	gen disapproval = Q7+Q8
	tab disapproval

* OTHER

* respondent's social class
	// note: there is no specific measure of class. using education as proxy for class as per protocol.
	tab EDUC
	* recoding 
	recode EDUC (1/4=8) (5=9) (7/8=11) (9=12) (10/11=14) (12=16) (13=18) (14=20)
	
* respondent's gender
	tab GENDER
	recode GENDER (1=0) (2=1), gen(woman)

********************************************************************************

* ANALYSIS


*Test-Hyp1: Disapproval towards pregnant women's alcohol consumption will vary by degree of consumption (i.e. coefficient of light drinking vs. no drinking will be smaller than coefficient of heavy drinking vs. no drinking)
	// note: if we use post-est test of light vs. no and heavy vs. no, we can't test the directional hyp.
	
	reg disapproval i.severity_drinking if severity_drinking!=0
	tess 2.severity_drinking +, init(CalarcoS38) 
	
	
*Test-Hyp2: Disapproval towards pregnant women's alcohol consumption will vary by women's social status (high vs. low)
	reg disapproval i.status_low##i.anydrinking 
	// reject. p=0.826
	tess 1.status_low#1.anydrinking
	
*Test-Hyp3: Disapproval towards pregnant women's alcohol consumption will vary by if signal about doctor's warning is included.
	reg disapproval i.doctorwarning##i.anydrinking
	// reject. p=0.328
	tess 1.doctorwarning#1.anydrinking
	
*Test-Hyp4: Disapproval towards pregnant women's alcohol consumption will be higher among less advantaged respondents.
	reg disapproval c.EDUC##i.anydrinking
	// reject. p=0.995
	tess EDUC#1.anydrinking
	
*Test-Hyp5: Disapproval towards pregnant women's alcohol consumption will vary among men and women respondents.
	reg disapproval i.woman##i.anydrinking
	// reject. p=0.090
	tess 1.woman#1.anydrinking
